Control of the semi-discrete 1D heat equation under nonnegative control constraint

Jérôme Lohéac[1], Emmanuel Trélat[2], and Enrique Zuazua[3]

[1]Institut de Recherche en Communications et Cyberntique de Nantes
[2]Universit Pierre et Marie Curie
[3]University of Deusto

April 9, 2018

1 Introduction

In the post IpOpt and AMPL use to solve time optimal control problems, we explain how to use `IpOpt` and `AMPL` in order to solve control problems with control constraints and possibly some state constraints.

In the present post, we are going to present a numerical development in order to find the minimal controllability time for the discretized heat equation with unilateral (non-negative) control constraint. More precisely, we consider the controllability of a discretized version of the 1D heat equation under nonnegative Dirichlet control constraint. The infinite dimensional system, we consider is

$$\begin{aligned} \dot{\psi}(t,x) &= \partial_x^2\psi(t,x) && (t>0,\, x\in(0,1)),\\ \partial_x\psi(t,0) &= 0 && (t>0),\\ \psi(t,1) &= u(t) && (t>0),\\ \psi(0,x) &= \psi^0(x) && (x\in(0,1)), \end{aligned}$$

where ψ is the state to control ψ^0 the initial state and u the Dirichlet control. The continuous version of this problem has been analysed in [2] and it has been shown that for every initial state $\psi^0 \in L^2(0,1)$ and every positive constant target ψ^1, the controllability of this system under the control constraint

$$u(t) \geqslant 0 \qquad (t \geqslant 0), \tag{1}$$

requires a positive waiting time as soon as the target state ψ^1 is different from the initial state ψ^0.

In this post, we consider a spatially discretized version with has the form

$$\dot{y}(t) = Ay(t) + Bu(t) \qquad (t>0), \tag{2}$$

where the matrices A and B are

$$A = (n+1)^2 \begin{pmatrix} -2 & 2 & 0 & \dots & \dots & 0 \\ 1 & -2 & 1 & 0 & \dots & 0 \\ 0 & \ddots & \ddots & \ddots & \ddots & \vdots \\ \vdots & \ddots & \ddots & \ddots & \ddots & 0 \\ \vdots & & \ddots & \ddots & \ddots & 1 \\ 0 & \dots & \dots & 0 & 1 & -2 \end{pmatrix} \in M_n(\mathbb{R}) \quad \text{and} \quad B = (n+1)^2 \begin{pmatrix} 0 \\ \vdots \\ \vdots \\ \vdots \\ 0 \\ 1 \end{pmatrix} \in M_{n,1}(\mathbb{R}). \quad (3)$$

where $n+1 > 2$ is the number of discretization points and $[y]_i(t)$ (the i^{th} component of $y(t) \in \mathbb{R}^n$) stands for $\psi(t,(i-1)/n)$. To obtain the above finite dimensional system, we have used centered finite differences.

The aim is then to minimize the time T such that there exist a control $u : [0,T] \to \mathbb{R}_+$ steering the solution of (2) from $y^0 \in \mathbb{R}^n$ to $y^1 \in \mathbb{R}^n$ in time T. For the sake of simplicity, we assume that y^0 and y^1 are constant vectors of $\mathbb{R}^n$, that is to say that

$$y^0 = u^0 \begin{pmatrix} 1 \\ \vdots \\ 1 \end{pmatrix} \in \mathbb{R}^n \quad \text{and} \quad y^1 = u^1 \begin{pmatrix} 1 \\ \vdots \\ 1 \end{pmatrix} \in \mathbb{R}^n, \quad (4)$$

for some $u^0 \in \mathbb{R}$ and $u^1 \in \mathbb{R}$ Since the control u shall satisfy the constraint (1), it is easy to see, using a discrete version of the comparison principle, that in order to have the existence of a solution, then we need $u^1 > 0$.

Note that, using again the comparison principle, we can also show that if $u^0 \geqslant 0$ then, whatever the control $u \geqslant 0$ is, the solution of (2) always satisfies $y(t) \geqslant 0$.

To conclude, the optimization problem we aim to solve is

$$\begin{array}{l} \inf \quad T \\ \left| \begin{array}{l} T \geqslant 0, \\ y(T;u;y^0) = y^1, \\ u \in L^\infty(0,T), \quad u(t) \geqslant 0 \quad (t \in [0,T] \text{ a.e.}), \end{array} \right. \end{array} \quad (5)$$

where y^0 and y^1 are given by (4) for some $u^0 \in \mathbb{R}$ and $u^1 \in \mathbb{R}_+^*$, and where $y(t;u;y^0)$ stands for the solution of (2) at time t, with control u and initial condition y^0.

In this post, in order to find numerically the minimal controllability time and a time optimal control, we are going to use the abstract result presented in the next paragraph. An other way could be to introduce, in addition to the control constraint $u(t) \geqslant 0$, the additional control constraint $u(t) \leqslant M$, and let M increase to $+\infty$. Such an approach has been introduced in the previous post IpOpt and AMPL use to solve time optimal control problems.

2 Abstract result

It is shown in [1] that at the minimal time $\underline{T}(y^0, y^1)$, defined by (5), there exists a non-negative control u in the class of Radon measure. Furthermore, it is proved in this article that at the minimal controllability time, there exist one and only one non-negative Radon measure and this time optimal control is a convex sum of at most $\lfloor (n+1)/2 \rfloor$ Dirac masses. In other words, the non-negative time optimal control take the form:

$$u = \sum_{k=1}^{N} \alpha_k \delta_{t_k},$$

with $N = \lfloor (n+1)/2 \rfloor$, $0 \leqslant \alpha_k$ and $0 \leqslant t_1 \leqslant \cdots \leqslant t_N \leqslant T$.
For a control of this form, the solution of (2) at time T, with initial condition y^0, is:

$$y(T) = e^{TA} y^0 + \sum_{k=1}^{N} \alpha_i e^{(T-t_k)A} B.$$

Consequently, the minimal time $\underline{T}(y^0, y^1)$ given by (5) is a minimizer of:

$$\inf \quad T \qquad \left| \begin{array}{l} 0 \leqslant t_1 \leqslant \cdots \leqslant t_N \leqslant T, \\ \alpha_1, \ldots, \alpha_N \geqslant 0, \\ y^1 - e^{TA} y^0 = \displaystyle\sum_{k=1}^{N} \alpha_k e^{(T-t_k)A} B. \end{array} \right. \tag{6}$$

3 Numerical implementation

The optimization problem (6) is not easy to solve directly, mainly because of the presence of a matrix exponential. Thus, instead we will solve the discretized heat equation on each time interval (t_k, t_{k+1}), with $t_0 = 0$ and $t_{N+1} = T$. Let us then set $\tau_k = t_{k+1} - t_k$ for every $k \in \{0, \ldots, N\}$. We have $T = \sum_{k=0}^{N} \tau_k$ and the optimization problem (6) also writes

$$\inf \quad \sum_{k=0}^{N} \tau_k \qquad \left| \begin{array}{l} 0 \leqslant \tau_k \qquad (k \in \{0, \ldots, N\}), \\ 0 \leqslant \alpha_k \qquad (k \in \{1, \ldots, N\}), \\ y_0(0) = y^0, \\ y_{k+1}(0) = y_k(\tau_k) + \alpha_{k+1} B \qquad (k \in \{0, \ldots, N-1\}), \\ y_N(\tau_N) = y^1, \\ \qquad \text{where } y_k(t) = e^{tA} y_k(0), \text{ i.e. } y_k \text{ is solution of (2) with null control.} \end{array} \right.$$

In the above constraints, $y_k(\tau) = y(\tau + t_k)$ for every $\tau \in (0, \tau_k)$, where y is the solution of (2), with control $u = \sum_{k=1}^{N} \alpha_k \delta_{t_k}$ and initial condition $y(0) = y^0$. Notice that since α_k is only constrained to be non-negative, and since the vector B (given by (3)) is of the form $(0, \ldots, 0, b_n)^\top$, with $b_n \geqslant 0$, the constraint $y_{k+1}(0) = y_k(\tau_k) + \alpha_{k+1} B$ can be expressed as

$$[y_{k+1}]_n(0) \geqslant [y_k]_n(\tau_k) \qquad \text{and} \qquad [y_{k+1}]_i(0) = [y_k]_i(\tau_k) \qquad (i \in \{1, \ldots, n-1\}).$$

Consequently, the parameters $\alpha_1, \ldots, \alpha_N$ can be forgotten.

In order to numerically compute $y_k(t) = e^{tA} y_k(0)$ for $t \in [0, \tau_k]$, we are going to use the Crank-Nicholson method. More precisely, given $N_t \in \mathbb{N}^*$, we approximate $y_k(j\tau_k/N_t)$ by y_k^j, with y_k^j solution of

$$y_k^0 = y_k(0), \qquad \left(\mathrm{I}_n - \frac{\tau_k}{2N_t} A \right) y_k^{j+1} = \left(\mathrm{I}_n + \frac{\tau_k}{2N_t} A \right) y_k^j \qquad (j \in \{0, \ldots, N_t - 1\}).$$

All in all, the fully discretized optimization problem, is

$$\min \sum_{k=0}^{N} \tau_k, \tag{7}$$

subject to the constraints

$$0 \leqslant \tau_k \qquad (k \in \{0, \dots, N\}), \tag{8}$$

$$[y_0^0]_i = u^0 \qquad (i \in \{1, \dots, n\}), \tag{9}$$

$$[y_{k+1}^0]_j = [y_k^{N_t}]_j \qquad (k \in \{0, \dots, N-1\},\ j \in \{1, \dots, n\}), \tag{10}$$

$$[y_{k+1}^0]_n \geqslant [y_k^{N_t}]_n \qquad (k \in \{0, \dots, N-1\}), \tag{11}$$

$$\max_{i \in \{1,\dots,n\}} \left| [y_N^{N_t}]_i - u^1 \right| \leqslant \varepsilon, \tag{12}$$

$$\left(\mathrm{I}_n - \frac{\tau_k}{2N_t} A \right) y_k^{j+1} = \left(\mathrm{I}_n + \frac{\tau_k}{2N_t} A \right) y_k^j \qquad (k \in \{0, \dots, N\},\ j \in \{0, \dots, N_t - 1\}). \tag{13}$$

Note that instead of imposing $[y_N^{N_t}]_i = u^1$, we chose to relax this condition in the constraint (12). This choice has been maid, since we do not expect that the numerical solution exactly reaches the target y^1. In practice, we will take, $\varepsilon = 1/(n\,N_t)$.

In term of AMPL language, the above constrained optimization problem is

```
# define parameters
param n   = 20;              # number of spatial discretization points
param Nt  = 400;             # number of time discretization points
param N   = floor((n+1)/2);  # maximal number of control impulses
param eps= 1/(n*Nt);         # relaxation parameter ε in (12)
param u0  = 1;               # parameters u^0 and u^1, see (4)
param u1  = 5;

# define variables
var y   {k in 0..N, j in 0..Nt, i in 1..n}; # unknowns [y_k^j]_i
var tau {k in 0..N} >=0;     # unknown impulse times τ_k subject to (8)

# objective function, see (7)
minimize T:
  sum {k in 0..N} tau[k];

# define the constraints
# constraint (13) for interior points
subject to y_dyn {k in 0..N, j in 1..Nt, i in 2..n-1}:
  y[k,j,i]-y[k,j-1,i] =
    (n+1)^2*(y[k,j,i-1]-2*y[k,j,i]+y[k,j,i+1]+y[k,j-1,i-1]-2*y[k,j-1,i]+y[k,j-1,i
    +1])/2*tau[k]/Nt;
# constraint (13) for i=n
subject to y_dyn_Dirichlet {k in 0..N, j in 1..Nt}:
  y[k,j,n]-y[k,j-1,n] =
    (n+1)^2*(y[k,j,n-1]-2*y[k,j,n]+y[k,j-1,n-1]-2*y[k,j-1,n])/2*tau[k]/Nt;
# constraint (13) for i=1
subject to y_dyn_Neuman {k in 0..N, j in 1..Nt}:
  y[k,j,1]-y[k,j-1,1] =
    (n+1)^2*(y[k,j,2]-y[k,j,1]+y[k,j-1,2]-y[k,j-1,1])/2*tau[k]/Nt;
# initial condition, constraint (9)
subject to y_init {i in 1..n}:
  y[0,0,i] = u0;
# terminal condition, constraint (12)
subject to y_end1 {i in 1..n}:
  y[N,Nt,i] >= u1-eps;
subject to y_end2 {i in 1..n}:
  y[N,Nt,i] <= u1+eps;
# constraint (10)
subject to continuity {k in 0..N-1, i in 1..n-1}:
  y[k+1,0,i]=y[k,Nt,i];
```

```
# constraint (11)
subject to jump {k in 0..N-1}:
  y[k+1,0,n]>=y[k,Nt,n];

# solve the problem with IpOpt
option solver ipopt;
option ipopt_options "max_iter=100000 linear_solver=mumps halt_on_ampl_error yes";
solve;

# display parameters and solution
printf: "u0 = %24.16e\n", u0;
printf: "u1 = %24.16e\n", u1;
printf: "T  = %24.16e\n", T;
printf: "n  = %d\n", n;
printf: "Nt = %d\n", Nt;
printf: "N  = %d\n", N;
printf: "Data\n";
printf: "t_k =";
printf {k in 0..N}:  " %24.16e\t", tau[k];
printf: "\n";
printf: "y =\n";
for {k in 0..N} {
  printf "k = %d\n", k;
  printf {j in 0..Nt, i in 1..n}: " %24.16e\n", y[k,j,i];
}

end; # quit AMPL
```

In the following simulations, we take $n = 20$ and $N_t = 400$. By post-treatment of the results (see the `Scilab` file in the post IpOpt and AMPL use to solve time optimal control problems for an example), we obtain the following results:

- for $u^0 = 1$ and $u^1 = 5$: The minimal time obtained is `0.1689856`.
  The control and state trajectories are displayed on Figures 1, 2 and 3.



Figure 1: Time optimal control evolution in order to steer $y^0 \equiv 1$ to $y^1 \equiv 5$. The minimal time computed is `0.1689856`. Dirac masses are represented by arrows. The corresponding state trajectory is given in Figres 2 and 3.



Figure 2: Evolution of the state between two Dirac impulses. The corresponding control required to steer $y^0 \equiv 1$ to $y^1 \equiv 5$ is given in Figure 1 and the minimal time computed is `0.1668338`. (See Figures 3 for the final times.)



Figure 3: Figure 2 continued.

- for $u^0 = 5$ and $u^1 = 1$: The minimal time obtained is `0.7267605`.
  The control and state trajectories are displayed on Figures 4, 5 and 6.



Figure 4: Time optimal control evolution in order to steer $y^0 \equiv 1$ to $y^1 \equiv 5$. The minimal time computed is `0.7267605`. Dirac masses are represented by arrows. The corresponding state trajectory is given in Figures 5 and 6.



Figure 5: Evolution of the state between two Dirac impulses. The corresponding control required to steer $y^0 \equiv 1$ to $y^1 \equiv 5$ is given in Figure 4 and the minimal time computed is `0.7267605`. (See Figure 6 for the final times.)



Figure 6: Figure 5 continued.

The videos corresponding to the two above simulations are available on the page:
• Control of the semi-discrete 1D heat equation under nonnegative control constraint.

References

[1] J. Lohéac, E. Trélat, and E. Zuazua. Control of the semi-discrete 1d heat equation under nonnegative control constraint. In preparation.

[2] J. Lohéac, E. Trélat, and E. Zuazua. Minimal controllability time for the heat equation under unilateral state or control constraints. *Mathematical Models and Methods in Applied Sciences*, 27(09):1587–1644, 2017.